Mètodes Algebraics per a l'Estadística

Grau d'Estadística Aplicada

1 Sumes i interseccions de subespais vectorials.

La suma i la intersecció de subespais vectorials donats és una de les situacions naturals en les que es *fabriquen* espais vectorials nous a partir dels que ja es tenen. El problema bàsic en aquestes situacions és determinar bases d'aquests *productes* a partir de les dades que es tinguin dels espais originals. Si es treballa amb **SageMath** i es tenen definits un parell d'espais vectorials V i W, la instrucció intesection( ) genera $V \cap W$, mentre que com en molts altres casos *l'operador suma* $+$ permet obtenir $V + W$ de forma immediata. D'aquesta forma els resultats són, també, objectes de la categoria *espai vectorial* i, pert tant, es disposa de tota la bateria de funcions associades (dimensió, base,...)

Exemple: Considerem el subespai vectorial F generat per $\vec{v}_1 = (1, 1, -1)$, $\vec{v}_2 = (2, 0, -1)$ i el subespai vectorial G generat per $\vec{w}_1 = (1, 0, -1)$, $\vec{w}_2 = (2, 3, 0)$, $\vec{w}_3 = (4, 3, -2)$.

```
In [ ]: v1=vector(QQ,[1,1,-1])
        v2=vector(QQ,[2,0,-1])
        w1=vector(QQ,[1,0,-1])
        w2=vector(QQ,[2,3,0])
        w3=vector(QQ,[4,3,-2])
        F=span([v1,v2])
        G=span([w1,w2,w3])
        show(F)
        show(G)
```

Ara ja es pot construir la suma i la intersecció:

```
In [ ]: S=F+G
        show(S)
        show(S.dimension())
```

```
In [ ]: Intrs=F.intersection(G)
        show(Intrs.basis())
```

```
In [ ]: show(Intrs.dimension())
```

Exercici

Sigui E el subespai vectorial donat pels vectors $\vec{x} = (x_1, x_2, x_3, x_4, x_5)$ que compleixen

$$\left.\begin{aligned} -x_1 + x_2 + x_3 - 5\,x_5 &= 0 \\ -x_1 + x_3 - 2\,x_4 - 2\,x_5 &= 0 \\ x_2 + 2\,x_4 - 3\,x_5 &= 0 \\ x_1 + x_2 - x_3 + 4\,x_4 - x_5 &= 0 \end{aligned}\right\}$$

Sigui F el subespai vectorial generat pels vectors

$$\begin{aligned}
\vec{v}_1 &= (-5,\ 5,\ -5,\ -1,\ 1)\,, \\
\vec{v}_2 &= (-15,\ -15,\ 5,\ 9,\ 1)\,, \\
\vec{v}_3 &= (-5,\ 35,\ -25,\ -13,\ 3)\,, \\
\vec{v}_4 &= (-15,\ 1,\ -4,\ 3,\ 1)
\end{aligned}$$

Doneu una base de E i una de F. Quines són les dimensions de cada un dels espais?
Calculeu els espais $G = E \cap F$ i $S = E + F$.
Doneu bases de G i S.
Comproveu que

$$\dim(S) = \dim(E) + \dim(F) - \dim(G)$$

---

2 Canvis de coordenades.

Quan en un mateix espai vectorial es consideren bases diferents, es poden *assignar coordenades* a cada vector de l'espai, que depenen de la base que es prengui com a referència. La transformació que relaciona les coordenades d'un cert vector respecte dues bases diferents de l'espai on s'està treballant queda determinada per la matriu construida amb les coordenades dels vectors d'una de les bases en funció de l'altra. Determinar les coordenades d'un vector respecte una base donada i calcular aquestes matrius de *canvi de base* són problemes que es poden resoldre, de forma més o menys immediata, utilitzant les intruccions de **SageMath** adequades. La instrucció clau en aquests problemes és subspace_with_basis( ). Aquesta funció permet definir un subespai vectorial explicitant una base concreta amb la que es vol treballar i, a més, es pot combinar amb la funció coordinates( ) que calcula les coordenades d'un vector respecte la base que tingui assignada l'espai vectorial donat.

Exemple: Considerem l'espai E generat pels vectors independents (base) $\vec{v}_1 = (1, -1, 1)$, $\vec{v}_2 = (2, 3, -1)$.

Comprovem en primer lloc que el vector $\vec{v} = (4, 1, 1)$ és un vector de E

In [ ]:
```
v1=vector(QQ,[1,-1,1])
v2=vector(QQ,[2,3,-1])
E=span(QQ,[v1,v2])
Ev=(QQ^3).subspace_with_basis([v1,v2])
show(E)
show(Ev)
show("Son iguals?")
show(E==Ev)
```

In [ ]:
```
v=vector(QQ,[4,1,1])
show(v)
```

In [ ]:
```
# Es pot comprovar que v
# es de E
v in E
```

```
In [ ]: # O fer-ho en la versió
        # Ev que genera el mateix espai
        v in Ev
```

Ara es poden calcular tant les coordenades de $\vec{v}$ respecte la base $\vec{v}_1$, $\vec{v}_2$ (Ev) com les que corresponen a la *base reduida* que surt en definir l'espai amb un span( ) (E).

```
In [ ]: Ev.coordinates(v)
```

```
In [ ]: 2*v1+v2,v
```

```
In [ ]: E.coordinates(v)
```

```
In [ ]: E.0
```

```
In [ ]: E.1
```

```
In [ ]: 4*E.0+E.1,v
```

Exercicis

Comproveu en primer lloc que els vectors $\vec{v}_1 = (1, 2, 1)$, $\vec{v}_2 = (2, 1, 2)$ i $\vec{w}_1 = (1, -1, 1)$, $\vec{w}_2 = (3, 1, 3)$ generen (i són bases) del mateix espai vectorial E.

Calculeu les coordenades de $\vec{v}_1$ i $\vec{v}_2$ respecte la base $\vec{w}_1, \vec{w}_2$

Tenint en compte que els coeficients que acabeu de calcular són les *columnes* de la matriu de canvi de coordenades, determineu la matriu M que permet obtenir les coordenades d'un vector qualsevol de E respecte la base $\vec{w}_1, \vec{w}_2$ en funció de les seves coordenades respecte la base $\vec{v}_1, \vec{v}_2$.

Comproveu que els càlculs estan ben fets obtenint les coordenades del vector de E donat per $\vec{x} = -3\,\vec{v}_1 + 2\,\vec{v}_2$ respecte la base $\vec{w}_1$, $\vec{w}_2$ fent la multiplicació de matrius adequada i comparant el resultat amb el que apareix si ho feu directament.

Comproveu que els vectors

$$\begin{aligned} \vec{v}_1 &= (0,\ 4,\ 1,\ 1)\,, \\ \vec{v}_2 &= (-1,\ -2,\ -4,\ 2)\,, \\ \vec{v}_3 &= (0,\ 4,\ 1,\ 2)\,, \\ \vec{v}_4 &= (1,\ -1,\ 3,\ -2) \end{aligned}$$

Determinen una base (de l'espai total de dimensió 4).

Calculeu la matriu de canvi de base M que permet obtenir les coordenades d'un vector qualsevol respecte aquesta base a partir de les seves components en la base canònica.

Calculeu les coordenades del vector que s'expressa canònicament com $\vec{x} = (5, -1, 2, 1)$ respecte la base $\vec{v}_1, \vec{v}_2, \vec{v}_3, \vec{v}_4$.

Considereu l'aplicació lineal que, respecte les bases canòniques dels espais, ve determinada per la matriu

$$A = \begin{pmatrix} 4 & 2 & 2 & -2 \\ -1 & 2 & 0 & -1 \\ -2 & -1 & 4 & 4 \end{pmatrix}$$

Si a l'espai de sortida es considera la base determinada pels vectors

$$\begin{aligned} \vec{v}_1 &= (-2,\, 1,\, -1,\, -2)\,, \\ \vec{v}_2 &= (-1,\, 0,\, -1,\, -1)\,, \\ \vec{v}_3 &= (0,\, -2,\, -1,\, 2)\,, \\ \vec{v}_4 &= (-1,\, 1,\, -1,\, -2) \end{aligned}$$

i en el d'arribada la que donen

$$\begin{aligned} \vec{w}_1 &= (-2,\, 1,\, -2)\,, \\ \vec{w}_2 &= (0,\, 1,\, -3)\,, \\ \vec{w}_3 &= (3,\, -2,\, 4) \end{aligned}$$

quina és la matriu de l'aplicació lineal respecte aquestes bases?