

Àlgebra Lineal

Grau d'Estadística Aplicada

1 Polinomi característic.

La funció de [SageMath](#) que calcula el polinomi característic d'una matriu és `charpoly` (o `characteristic_polynomial` si no es té mandra d'escriure una mica més). Si es vol calcular el polinomi característic de la matriu

$$A = \begin{pmatrix} 3 & -6 & -8 & 6 \\ 2 & -3 & -4 & 4 \\ -2 & 3 & 3 & -3 \\ -2 & 1 & -2 & 0 \end{pmatrix}$$

només s'haurà de fer:

```
In [ ]: A=matrix(QQ,4,4,[3, -6, -8, 6,
                        2, -3, -4, 4,
                        -2, 3, 3, -3,
                        -2, 1, -2, 0])

show(A)
```

```
In [ ]: show(A.charpoly())
```

Podeu veure que el resultat coincideix amb el que apareix si calculem directament utilitzant determinants.

```
In [ ]: show((A-x*identity_matrix(4)).det())
```

Si es vol utilitzar algun altre símbol com indeterminada del polinomi (hi podria haver conflictes si hi ha paràmetres simbòlics entre els coeficients de la matriu), es pot especificar en els arguments:

```
In [ ]: show(A.charpoly('t'))
```

```
In [ ]: B=matrix(4,4,[x, -6, x^2, 6,
                    2, -3, -4, 4,
                    -2, 3, x-1, -3,
                    -2, 1, -2, 0])

show(B)
```

```
In [ ]: show(B.charpoly()) # La x té dos significats diferents!!!!
```

```
In [ ]: show(B.charpoly('lambda')) # D'aquesta forma desapareix la indeterminació
```

Observació tècnica: El resultat de la funció `charpoly` és un objecte de la categoria *polinomi* que és un entremig d'un objecte amb variables simbòliques i una funció. Si proveu d'aplicar a aquest tipus d'objecte algunes de les funcions que esteu acostumats a utilitzar (com, per exemple, `solve` o `expand`) veureu que poden aparèixer alguns problemes.

2 Valors propis

La instrucció per obtenir els valors propis d'una matriu és `eigenvalues()`. El resultat és una llista d'aquests valors propis tenint en compte la multiplicitat.

```
In [ ]: show(A.eigenvalues())
```

Es pot veure com el valor 1 és *doblet* fent la factorització del polinomi característic.

```
In [ ]: show(A.charpoly().factor())
```

Com que els valors propis són les solucions d'una equació polinòmica, el *normal* és que siguin **números complexos** amb parts real i imaginària no entera. Per tant, si feu proves amb matrius a l'atzar, en general veureu uns resultats *complicats*.

3 Vectors propis

Cada valor propi λ de la matriu A té associat el subespai vectorial dels *vectors propis* corresponents. Que són els vectors \vec{v} que compleixen $A \cdot \vec{v} = \lambda \vec{v}$. Poden obtenir aquestes dades amb `eigenspaces_right()` o `eigenvectors_right()`.

```
In [ ]: show(A.eigenspaces_right())
```

```
In [ ]: show(A.eigenvectors_right())
```

Noteu que en el primer cas el resultat és una llista de parells formats per un dels *valors propis* i el *subespai dels vectors propis associat*. En canvi, en el segon exemple, el resultat és una llista de tripletes formades pel valor propi, una base de l'espai de vectors propis associats i la multiplicitat.

Exercicis

Calculeu, per a cada una de les matrius següents, el polinomi característic, els valors propis i els espais de vectors propis corresponents a cada un d'ells (donant una base de cada un d'ells).

$$A1 = \begin{pmatrix} -8 & -4 & 24 \\ 8 & 5 & -20 \\ -2 & -1 & 6 \end{pmatrix}$$

$$A2 = \begin{pmatrix} -2 & -1 & -2 & 1 \\ 1 & 1 & 1 & 0 \\ -1 & -3 & 3 & -1 \\ -2 & -5 & 3 & -2 \end{pmatrix}$$

$$A3 = \begin{pmatrix} 7 & 3 & 2 & 18 & 12 \\ -2 & 0 & 0 & -4 & -4 \\ 5 & 3 & 2 & 12 & 10 \\ -1 & -1 & 0 & 0 & -2 \\ -2 & 0 & -1 & -9 & -3 \end{pmatrix}$$

$$A4 = \begin{pmatrix} -4 & 2 & -4 & -2 \\ -7 & 4 & -5 & -2 \\ 1 & -2 & -5 & -4 \\ 1 & 2 & 11 & 8 \end{pmatrix}$$

$$A5 = \begin{pmatrix} 5 & 2 & 3 & 1 \\ -2 & 0 & -1 & 0 \\ -6 & -3 & -4 & -2 \\ 2 & 0 & 1 & 0 \end{pmatrix}$$

$$A6 = \begin{pmatrix} -1 & -3 & -2 & -1 \\ 4 & 9 & 4 & 4 \\ -4 & -9 & -3 & -5 \\ -4 & -8 & -4 & -3 \end{pmatrix}$$

4 Diagonalització

Amb els càlculs que surten en els exercicis anteriors ja és possible determinar quan una matriu és diagonalitzable o no. No obstant, també hi ha funcions específiques (basades en les que ja heu utilitzat) per a decidir directament si una matriu és diagonalitzable i, fins i tot, obtenir la matriu de canvi de base.

Es pot saber si una matriu és diagonalitzable amb el *test* `is_diagonalizable()` (dona com a resultat **cert** o **fals**) i s'obté la forma diagonal i la matriu de canvi de base amb la instrucció `eigenmatrix_right()` (quan s'aplica s'obté una llista amb aquests dos elements).

Amb la matriu A que hi ha definida des del principi (que és diagonalitzable com ja deveu haver vist) tindrem:

```
In [ ]: A.is_diagonalizable()
```

```
In [ ]: [D,M]=A.eigenmatrix_right()
        show(D)
        show(M)
```

```
In [ ]: show(M.inverse()*A*M)
        show(D)
```

```
In [ ]: show(M*D*M.inverse())  
        show(A)
```

Exercicis

Determineu quines de les matrius de l'apartat anterior són diagonalitzables. Doneu la matriu de canvi de base en aquests casos.

Considereu la matriu

$$A = \begin{pmatrix} 4 & -6 & -6 \\ -3 & 7 & 6 \\ 3 & -6 & -5 \end{pmatrix}$$

Comproveu que és diagonalitzable i calculeu matrius M (invertible) i D (diagonal) tals que $M^{-1} \cdot A \cdot M = D$

Utilitzant les dades obtingudes a l'apartat anterior, determineu **totes** les matrius R que compleixen $R^2 = A$ que sigueu capaços.